ERIC BAILEY

PAIP EXERCISES

Contents

Introduction to Common Lisp

Using Functions

1 ⟨*titles* 1⟩≡ (7)

```
(defparameter *titles*
  '(Mr Mrs Miss Ms Sir Madam Dr Admiral Major General)
  "A list of titles that can appear at the start of a name.")
```

Defines:
titles, used in chunk 4.

2 ⟨*abstract first-name* 2⟩≡

```
(⟨function first-name(name): 3⟩
  (⟨if the first element of name is a title 4⟩
    ⟨then return the first-name of the rest of the name 5⟩
    ⟨else return the first element of the name 6⟩))
```

3 ⟨*function first-name(name):* 3⟩≡ (2)

```
defun first-name (name)
 "Select the first name from a name represented as a list."
```

4 ⟨*if the first element of name is a title* 4⟩≡ (2)

```
if (member (first name) *titles*)
```

Uses *titles* 1.

5 ⟨*then return the first-name of the rest of the name* 5⟩≡ (2)

```
    (first-name (rest name))
```

6 ⟨*else return the first element of the name* 6⟩≡ (2)

```
(first name)
```

Exercises

7 ⟨*src/intro.lisp* 7⟩≡

```
(in-package #:paip)
(defpackage #:paip.intro
  (:use #:cl #:lisp-unit))
(in-package #:paip.intro)

⟨titles 1⟩

;; Exercise 1.1
⟨Exercise 1.1 8⟩

;; Exercise 1.2
⟨Exercise 1.2 17⟩

;; Exercise 1.3
⟨Exercise 1.3 24⟩

;; Exercise 1.4
⟨Exercise 1.4 28⟩

;; Exercise 1.5
⟨Exercise 1.5 33⟩
```

Uses use 54.

Exercise 1.1

Define a version of last-name that handles "Rex Morgan MD," "Morton Downey, Jr.," and whatever other cases you can think of.

8 ⟨*Exercise 1.1* 8⟩≡ (7)

```
⟨suffixes 13⟩

⟨last-name 9⟩

⟨Exercise 1.1 tests 14⟩
```

9 ⟨*last-name* 9⟩≡ (8)

```
(defun last-name (name)
  "Select the last name from a name represented as a list."
  (if ⟨the last element of a name is a suffix 10⟩
      ⟨then return the last-name of all but the last element of the name 11⟩
    ⟨else return the last element of the name 12⟩))
```

Defines:
last-name, used in chunks 11, 15, and 16.

First, we check to see if the last element of the name is a suffix, i.e. whether it's a member of *suffixes*.

10 ⟨*the last element of a name is a suffix* 10⟩≡ (9)

```
(member (first (last name)) *suffixes*)
```

Uses *suffixes* 13.

If it is, then drop it from the `name` and return the `last-name` of the result.

11 ⟨*then return the* `last-name` *of all but the last element of the name* 11⟩≡ (9)
```
(last-name (butlast name))
```
Uses `last-name` 9.

Otherwise, it's the last name, so return it.

12 ⟨*else return the last element of the name* 12⟩≡ (9)
```
(first (last name))
```

Define some well-known suffixes.

13 ⟨*suffixes* 13⟩≡ (8)
```
(defparameter *suffixes*
  '(MD Jr. Sr. III)
  "A list of suffixes that can appear at the end of a name.")
```
Defines:
`*suffixes*`, used in chunk 10.

14 ⟨*Exercise 1.1 tests* 14⟩≡ (8)
```
(define-test test-last-name
  ⟨Rex Morgan MD 15⟩
  ⟨Morton Downey, Jr. 16⟩)
```

15 ⟨*Rex Morgan MD* 15⟩≡ (14)
```
(assert-equal 'Morgan (last-name '(Rex Morgan MD)))
```
Uses `last-name` 9.

Assert that the `last-name` of *Rex Morgan MD* is *Morgan*.

16 ⟨*Morton Downey, Jr.* 16⟩≡ (14)
```
(assert-equal 'Downey (last-name '(Morton Downey Jr.)))
```
Uses `last-name` 9.

Exercise 1.2

Write a function to exponentiate, or raise a number to an integer power. For example `(power 3 2)` $= 3^2 = 9$.

17 ⟨*Exercise 1.2* 17⟩≡ (7)
```
⟨square 22⟩

⟨power 18⟩

⟨Exercise 1.2 tests 23⟩
```

18 ⟨*power* 18⟩≡ (17)
```
(defun power (x n)
  "Raise x to the power of n."
  (cond ⟨if n is zero return 1 19⟩
        ⟨if n is even return x to the power of n over two, squared 20⟩
        ⟨otherwise return x times x to the power of n minus one 21⟩))
```
Defines:
`power`, used in chunks 20, 21, and 23.

$$x^n = \begin{cases} 1 & \text{if } n = 0, \\ (x^{n/2})^2 & \text{if } n \text{ is even}, \\ x \times x^{n-1} & \text{otherwise.} \end{cases}$$

19 ⟨*if n is zero return 1* 19⟩≡ (18) $x^0 = 1$

```
((zerop n) 1)
```

20 ⟨*if n is even return x to the power of n over two, squared* 20⟩≡ (18)

```
((evenp n) (square (power x (/ n 2))))
```

Uses `power` 18 and `square` 22.

21 ⟨*otherwise return x times x to the power of n minus one* 21⟩≡ (18)

```
(t (* x (power x (- n 1))))
```

Uses `power` 18.

22 ⟨*square* 22⟩≡ (17) $\text{square}(x) = x^2$

```
(defun square (x) (expt x 2))
```

Defines:
`square`, used in chunk 20.

23 ⟨*Exercise 1.2 tests* 23⟩≡ (17)

```
(define-test test-power
  (assert-equal 9 (power 3 2)))
```

Uses `power` 18.

Exercise 1.3

Write a function that counts the number of atoms in an expression. For example: `(count-atoms '(a (b) c))` = 3. Notice that there is something of an ambiguity in this: should `(a nil c)` count as three atoms, or as two, because it is equivalent to `(a () c)`?

24 ⟨*Exercise 1.3* 24⟩≡ (7)

```
(defun count-atoms (exp &optional (if-null 1))
  "Return the total number of atoms in the expression,
   counting nil as an atom only in non-tail position."
  (cond ⟨if exp is nil there are if-null atoms 25⟩
        ⟨if exp is an atom there is only one 26⟩
        ⟨otherwise add the count of the atoms in the first and rest of exp 27⟩))
```

Defines:
`count-atoms`, used in chunk 27.

25 ⟨*if exp is nil there are if-null atoms* 25⟩≡ (24)

```
((null exp) if-null)
```

26 ⟨*if exp is an atom there is only one* 26⟩≡ (24)

```
((atom exp) 1)
```

27 ⟨*otherwise add the count of the atoms in the first and rest of exp* 27⟩≡ (24)

```
(t (+ (count-atoms (first exp) 1)
      (count-atoms (rest exp) 0)))
```

Uses `count-atoms` 24.

Exercise 1.4

28 ⟨*Exercise 1.4* 28⟩≡ (7)

```
(defun count-anywhere (item tree)
  "Count the occurrences of item anywhere within tree."
  (cond ⟨if item is equal to tree, there is one occurrence 29⟩
        ⟨if tree is an atom, there are no occurrences 30⟩
        ⟨otherwise, add the occurrence within first the first and rest of tree 31⟩))

⟨Exercise 1.4 tests 32⟩
```

Defines:
`count-anywhere`, used in chunks 31 and 32.

29 ⟨*if item is equal to tree, there is one occurrence* 29⟩≡ (28)

```
((eql item tree) 1)
```

30 ⟨*if tree is an atom, there are no occurrences* 30⟩≡ (28)

```
((atom tree) 0)
```

31 ⟨*otherwise, add the occurrence within first the first and rest of tree* 31⟩≡ (28)

```
(t (+ (count-anywhere item (first tree))
      (count-anywhere item (rest tree))))
```

Uses `count-anywhere` 28.

32 ⟨*Exercise 1.4 tests* 32⟩≡ (28)

```
(define-test test-count-anywhere
  (assert-equal 3 (count-anywhere 'a '(a ((a) b) a))))
```

Uses `count-anywhere` 28.

Exercise 1.5

33 ⟨*Exercise 1.5* 33⟩≡ (7)

```
(defun dot-product (lhs rhs)
  "Compute the mathematical dot product of two vectors."
  ⟨multiply elements of the vectors pairwise and sum the results 34⟩)

⟨Exercise 1.5 tests 35⟩
```

Defines:
`dot-product`, used in chunk 35.

34 ⟨*multiply elements of the vectors pairwise and sum the results* 34⟩≡ (33)

```
(apply #'+ (mapcar #'* lhs rhs))
```

35 ⟨*Exercise 1.5 tests* 35⟩≡ (33)

```
(define-test test-dot-product
  (assert-equal 110 (dot-product '(10 20) '(3 4))))
```

Uses `dot-product` 33.

Higher-Order Functions

36 ⟨*mappend* 36⟩≡ (39 85)

```
(defun mappend (fn the-list)
  "Apply fn to each element of list and append the results."
  (apply #'append (mapcar fn the-list)))
```

Defines:
`mappend`, used in chunks 59 and 88.

A Simple Lisp Program

37 ⟨*random-elt* 37⟩≡ (85)

```
(defun random-elt (choices)
  "Choose an element from a list at random."
  (elt choices (random (length choices))))
```

Defines:
`random-elt`, used in chunk 86.

Overview of Lisp

38 ⟨*find-all* 38⟩≡ (39)

```
(defun find-all (item sequence &rest keyword-args
                 &key (test #'eql) test-not &allow-other-keys)
  "Find all those elements of sequence that match item,
  according to the keywords. Doesn't alter sequence."
  (if test-not
      (apply #'remove item sequence
             :test-not (complement test-not) keyword-args)
      (apply #'remove item sequence
             :test (complement test) keyword-args)))

;; (setf (symbol-function 'find-all-if) #'remove-if-not)
```

Defines:
`find-all`, used in chunk 44.

GPS: The General Problem Solver

39 ⟨*src/gps.lisp* 39⟩≡

```
(in-package #:paip)
(defpackage #:paip.gps
  (:use #:cl #:lisp-unit)
  (:shadow #:debug)
  (:export #:GPS))
(in-package #:paip.gps)
```

⟨*find-all* 38⟩

⟨*mappend* 36⟩

⟨*A list of available operators* 40⟩

⟨*An operation with preconds, add-list and del-list* 41⟩

⟨*Solve a goal from a state using a list of operators* 42⟩

⟨*Achieve an individual goal* 44⟩

⟨*Achieve all goals* 43⟩

⟨*Decide if an operator is appropriate for a goal* 45⟩

⟨*Apply operator to current state* 46⟩

⟨*Auxiliary Functions* 47⟩

⟨*Nursery School Example* 57⟩

⟨*Monkey and Bananas Example* 58⟩

⟨*The Maze Searching Domain* 59⟩

⟨*Maze Tests* 60⟩

⟨*Convert existing operators* 53⟩

⟨*The Blocks World Domain* 61⟩

⟨*Print debugging information* 65⟩

⟨*GPS Tests* 67⟩

Uses debug 65, GPS 42, and use 54.

40 ⟨*A list of available operators* 40⟩≡ (39)

```
(defvar *ops* nil "A list of available operators.")
```

Defines:
ops, used in chunks 42, 44, and 54.

41 ⟨*An operation with preconds, add-list and del-list* 41⟩≡ (39)

```
(defstruct op
  "An operation"
  (action nil)
  (preconds nil)
  (add-list nil)
  (del-list nil))
```

Uses op 52.

42 ⟨*Solve a goal from a state using a list of operators* 42⟩≡ (39)

```
(defun GPS (state goals &optional (*ops* *ops*))
  "General Problem Solver: from state, achieve goals using *ops*."
  (remove-if-not #'action-p
                 (achieve-all (cons '(start) state) goals nil)))
```

Defines:
GPS, used in chunks 39 and 59.
Uses *ops* 40, achieve 44, achieve-all 43, and action-p 49.

43 ⟨*Achieve all goals* 43⟩≡ (39)

```
(defun achieve-all (state goals goal-stack)
  "Achieve each goal, trying several orderings."
  (some #'(lambda (goals) (achieve-each state goals goal-stack))
        (orderings goals)))

(defun achieve-each (state goals goal-stack)
  "Try to achieve each goal, then make sure they still hold."
  (let ((current-state state))
    (if (and (every #'(lambda (g)
                        (setf current-state
                              (achieve current-state g goal-stack)))
                    goals)
             (subsetp goals current-state :test #'equal))
        current-state)))

(defun orderings (lst)
  (if (> (length lst) 1)
      (list lst (reverse lst))
      (list lst)))
```

Defines:
achieve-all, used in chunks 42 and 46.
achieve-each, never used.
orderings, never used.
Uses achieve 44.

44 ⟨*Achieve an individual goal* 44⟩≡ (39)

```
(defun achieve (state goal goal-stack)
  "A goal is achieved if it already holds,
  or if there is an appropriate op for it that is applicable."
  (dbg-indent :gps (length goal-stack) "Goal: ~a" goal)
  (cond ((member-equal goal state)      state)
        ((member-equal goal goal-stack) nil)
        (t (some #'(lambda (op) (apply-op state goal op goal-stack))
                 (appropriate-ops goal state)))))

(defun appropriate-ops (goal state)
  "Return a list of appropriate operators,
  sorted by the number of unfulfilled preconditions."
  (sort (copy-list (find-all goal *ops* :test #'appropriate-p)) #'<
        :key #'(lambda (op)
                 (count-if #'(lambda (precond)
                               (not (member-equal precond state)))
                           (op-preconds op)))))
```

Defines:
  `achieve`, used in chunks 42 and 43.
  `appropriate-ops`, never used.
Uses `*ops*` 40, `apply-op` 46, `appropriate-p` 45, `dbg-indent` 65, `find-all` 38, `member-equal` 55, and `op` 52.

45 ⟨*Decide if an operator is appropriate for a goal* 45⟩≡ (39)

```
(defun appropriate-p (goal op)
  "An op is appropriate to a goal if it is in its add list."
  (member-equal goal (op-add-list op)))
```

Defines:
  `appropriate-p`, used in chunk 44.
Uses `member-equal` 55 and `op` 52.

46 ⟨*Apply operator to current state* 46⟩≡ (39)

```
(defun apply-op (state goal op goal-stack)
  "Return a new, transformed state if op is applicable."
  (dbg-indent :gps (length goal-stack) "Consider: ~a" (op-action op))
  (let ((state* (achieve-all state (op-preconds op)
                             (cons goal goal-stack))))
    (unless (null state*)
      (dbg-indent :gps (length goal-stack) "Action: ~a" (op-action op))
      (append (remove-if #'(lambda (x)
                             (member-equal x (op-del-list op)))
                         state*)
              (op-add-list op)))))
```

Defines:
  `apply-op`, used in chunk 44.
Uses `achieve-all` 43, `dbg-indent` 65, `member-equal` 55, and `op` 52.

Auxiliary Functions

47 ⟨*Auxiliary Functions* 47⟩≡ (39)
⟨*Is a condition an executing form?* 48⟩

⟨*Is x an action?* 49⟩

⟨*Is the argument a list that starts with a given atom?* 50⟩

⟨*Convert an operator to use the executing convention* 51⟩

⟨*Create an operator* 52⟩

⟨*Use a list of of operators* 54⟩

⟨*Test if an element is equal to a member of a list* 55⟩

48 ⟨*Is a condition an executing form?* 48⟩≡ (47)
```
(defun executing-p (x)
  "Is x of the form: (executing ...) ?"
  (starts-with x 'executing))
```
Defines:
`executing-p`, used in chunks 49 and 51.
Uses `starts-with` 50.

49 ⟨*Is x an action?* 49⟩≡ (47)
```
(defun action-p (x)
  "Is x something that is (start) or (executing ...)?"
  (or (equal x '(start)) (executing-p x)))
```
Defines:
`action-p`, used in chunk 42.
Uses `executing-p` 48.

50 ⟨*Is the argument a list that starts with a given atom?* 50⟩≡ (47 85)
```
(defun starts-with (list x)
  "Is this a list whose first element is x?"
  (and (consp list) (eql (first list) x)))
```
Defines:
`starts-with`, used in chunks 48 and 78.

51 ⟨*Convert an operator to use the executing convention* 51⟩≡ (47)
```
(defun convert-op (op)
  "Make op conform to the (EXECUTING op) convention."
  (unless (some #'executing-p (op-add-list op))
    (push (list 'executing (op-action op)) (op-add-list op)))
  op)
```
Defines:
`convert-op`, used in chunks 52 and 53.
Uses `executing-p` 48 and `op` 52.

52 ⟨*Create an operator* 52⟩≡ (47)

```
(defun op (action &key preconds add-list del-list)
  "Make a new operator that obeys the (EXECUTING op) convention."
  (convert-op (make-op :action action
                       :preconds preconds
                       :add-list add-list
                       :del-list del-list)))
```

Defines:
op, used in chunks 41, 44–46, 51, 58, 59, and 61.
Uses convert-op 51.

53 ⟨*Convert existing operators* 53⟩≡ (39)

```
(mapc #'convert-op *school-ops*)
```

Uses *school-ops* 57 and convert-op 51.

54 ⟨*Use a list of of operators* 54⟩≡ (47)

```
(defun use (oplist)
  "Use oplist as the default list of operators."
  (length (setf *ops* oplist)))
```

Defines:
use, used in chunks 7, 39, 60, 62, 68, 91, and 92.
Uses *ops* 40.

55 ⟨*Test if an element is equal to a member of a list* 55⟩≡ (47)

```
(defun member-equal (item list)
  (member item list :test #'equal))
```

Defines:
member-equal, used in chunks 44–46.

Nursery School Example

To drive the son to school, the son must start at home and the car must work.

56 ⟨*Drive son to school* 56⟩≡ (57)

```
(make-op :action 'drive-son-to-school
         :preconds '(son-at-home car-works)
         :add-list '(son-at-school)
         :del-list '(son-at-home))
```

57 ⟨*Nursery School Example* 57⟩≡ (39)

```
(defparameter *school-ops*
  (list
   ⟨Drive son to school 56⟩
   (make-op :action 'shop-installs-battery
            :preconds '(car-needs-battery shop-knows-problem shop-has-money)
            :add-list '(car-works))
   (make-op :action 'tell-shop-problem
            :preconds '(in-communication-with-shop)
            :add-list '(shop-knows-problem))
   (make-op :action 'telephone-shop
            :preconds '(know-phone-number)
            :add-list '(in-communication-with-shop))
   (make-op :action 'look-up-number
            :preconds '(have-phone-book)
            :add-list '(know-phone-number))
   (make-op :action 'give-shop-money
            :preconds '(have-money)
            :add-list '(shop-has-money)
            :del-list '(have-money))))
```

Defines:
school-ops, used in chunks 53 and 67.

Monkey and Bananas

58 ⟨*Monkey and Bananas Example* 58⟩≡ (39)

```
(defparameter *banana-ops*
  (list
   (op 'climb-on-chair
       :preconds '(chair-at-middle-room at-middle-room on-floor)
       :add-list '(at-bananas on-chair)
       :del-list '(at-middle-room on-floor))
   (op 'push-chair-from-door-to-middle-room
       :preconds '(chair-at-door at-door)
       :add-list '(chair-at-middle-room at-middle-room)
       :del-list '(chair-at-door at-door))
   (op 'walk-from-door-to-middle-room
       :preconds '(at-door on-floor)
       :add-list '(at-middle-room)
       :del-list '(at-door))
   (op 'grasp-bananas
       :preconds '(at-bananas empty-handed)
       :add-list '(has-bananas)
       :del-list '(empty-handed))
   (op 'drop-ball
       :preconds '(has-ball)
       :add-list '(empty-handed)
       :del-list '(has-ball))
   (op 'eat-bananas
       :preconds '(has-bananas)
       :add-list '(empty-handed not-hungry)
       :del-list '(has-bananas hungry))))
```

Uses op 52.

The Maze Searching Domain

59 ⟨*The Maze Searching Domain* 59⟩≡ (39)

```
(defun make-maze-ops (pair)
  "Make maze ops in both directions."
  (list (make-maze-op (first pair) (second pair))
        (make-maze-op (second pair) (first pair))))

(defun make-maze-op (here there)
  "Make an operator to move between two places."
  (op `(move from ,here to ,there)
      :preconds `((at ,here))
      :add-list `((at ,there))
      :del-list `((at ,here))))

(defparameter *maze-ops*
  (mappend #'make-maze-ops
    '((1 2) (2 3) (3 4) (4 9) (9 14) (9 8) (8 7) (7 12) (12 13)
      (12 11) (11 6) (11 16) (16 17) (17 22) (21 22) (22 23)
      (23 18) (23 24) (24 19) (19 20) (20 15) (15 10) (10 5) (20 25))))

(defun find-path (start end)
  "Search a maze for a path from start to end."
  (let ((results (GPS `((at ,start)) `((at ,end)))))
    (unless (null results)
      (cons start (mapcar #'destination
                          (remove '(start) results
                                  :test #'equal))))))

(defun destination (action)
  "Find the Y in (executing (move from X to Y))."
  (fifth (second action)))
```

Defines:
`destination`, never used.
`find-path`, used in chunk 60.
`make-maze-op`, never used.
`make-maze-ops`, never used.
Uses `GPS` 42, `mappend` 36, and `op` 52.

Tests

60 ⟨*Maze Tests* 60⟩≡ (39)

```
(define-test maze
  (use *maze-ops*)
  (assert-equal '(1 2 3 4 9 8 7 12 11 16 17 22 23 24 19 20 25)
                (find-path 1 25)))

(define-test go-nowhere
  (use *maze-ops*)
  (assert-equal '(1) (find-path 1 1)))

(define-test maze-reverse
  (use *maze-ops*)
  (assert-equal (find-path 1 25) (reverse (find-path 25 1))))
```

Uses `find-path` 59 and `use` 54.

The moral is that when a programmer uses puns—saying what's convenient instead of what's really happening–there's bound to be trouble.

The Blocks World Domain

61 ⟨*The Blocks World Domain* 61⟩≡ (39)

```
(defun make-block-ops (blocks)
  (let ((ops nil))
    (dolist (a blocks)
      (dolist (b blocks)
        (unless (equal a b)
          (dolist (c blocks)
            (unless (or (equal c a)
                        (equal c b))
              (push (move-op a b c) ops)))
          (push (move-op a 'table b) ops)
          (push (move-op a b 'table) ops))))
    ops))

(defun move-op (a b c)
  "Make an operator to move A from B to C."
  (op `(move ,a from ,b to ,c)
      :preconds `((space on ,a) (space on ,c) (,a on ,b))
      :add-list (move-ons a b c)
      :del-list (move-ons a c b)))

(defun move-ons (a b c)
  (if (eq b 'table)
      `((,a on ,c))
      `((,a on ,c) (space on ,b))))
```

⟨*Blocks World Tests* 62⟩

Defines:
`make-block-ops`, used in chunk 62.
`move-ons`, never used.
`move-op`, never used.
Uses `op` 52.

62 ⟨*Blocks World Tests* 62⟩≡ (61)

```
(define-test simplest-blocks-problem
  (use (make-block-ops '(a b)))
  (assert-equal '((start) (executing (move a from table to b)))
                (gps '((a on table) (b on table) (space on a) (space on b)
                       (space on table))
                     '((a on b) (b on table)))))

(define-test slighty-more-complex-blocks
  (use (make-block-ops '(a b)))
  (assert-equal '((start)
                  (executing (move a from b to table))
                  (executing (move b from table to a)))
                (gps '((a on b) (b on table) (space on a) (space on table))
                     '((b on a)))))

(define-test blocks-goals-order-insignificant
  (let ((ops (make-block-ops '(a b c))))
    (let ((state '((a on b) (b on c) (c on table)
                   (space on a) (space on table))))
      (assert-equal '((start)
                      (executing (move a from b to table))
                      (executing (move b from c to a))
                      (executing (move c from table to b)))
                    (gps state '((b on a) (c on b)) ops))
      (assert-equal '((start)
                      (executing (move a from b to table))
                      (executing (move b from c to a))
                      (executing (move c from table to b)))
                    (gps state '((c on b) (b on a)) ops)))))

(define-test blocks-ops-ordered-intelligently
  (let ((ops (make-block-ops '(a b c))))
    (let ((state '((c on a) (a on table) (b on table)
                   (space on c) (space on b) (space on table))))
      (assert-equal '((start)
                      (executing (move c from a to table))
                      (executing (move a from table to b)))
                    (gps state '((c on table) (a on b)) ops)))
    (let ((state '((a on b) (b on c) (c on table)
                   (space on a) (space on table))))
      (assert-equal '((start)
                      (executing (move a from b to table))
                      (executing (move b from c to a))
                      (executing (move c from table to b)))
                    (gps state '((b on a) (c on b)) ops))
      (assert-equal '((start)
                      (executing (move a from b to table))
                      (executing (move b from c to a))
                      (executing (move c from table to b)))
                    (gps state '((c on b) (b on a)) ops)))))
```

⟨*Blocks: The Sussman Anomaly* 63⟩

Uses `make-block-ops` 61 and `use` 54.

The Sussman Anomaly

N.B. These results are undesirable and will be addressed in chapter 6.

63 ⟨*Blocks: The Sussman Anomaly* 63⟩≡ (62)

```
(define-test blocks-the-sussman-anomaly
  (let ((start '((c on a) (a on table) (b on table)
                 (space on c) (space on b) (space on table))))
    (assert-nil (gps start '((a on b) (b on c))))
    (assert-nil (gps start '((b on c) (a on b))))))
```

Debugging

64 ⟨*Debugging usage* 64⟩≡

```
;; Example call
(dbg :gps "The current goal is: ~a" goal)

;; Turn on debugging
(debug :gps)

;; Turn off debugging
(undebug :gps)
```

Uses `dbg` 65, `debug` 65, and `undebug` 65.

65 ⟨*Print debugging information* 65⟩≡ (39)

```
(defvar *dbg-ids* nil
  "Identifiers used by dbg")

(defun dbg (id format-string &rest args)
  "Print debugging info if (DEBUG ID) has been specified."
  (when (member id *dbg-ids*)
    (format *debug-io* "~&~?" format-string args)))

(defun debug (&rest ids)
  "Start dbg output on the given ids."
  (setf *dbg-ids* (union ids *dbg-ids*)))

(defun undebug (&rest ids)
  "Stop dbg on the ids. With no ids, stop dbg altogether."
  (setf *dbg-ids* (if (null ids) nil
                      (set-difference *dbg-ids* ids))))

(defun dbg-indent (id indent format-string &rest args)
  "Print indented debugging info if (DEBUG ID) has been specified."
  (when (member id *dbg-ids*)
    (format *debug-io* "~&~V@T~?" (* 2 indent) format-string args)))
```

Defines:
dbg-ids, never used.
dbg, used in chunk 64.
dbg-indent, used in chunks 44 and 46.
debug, used in chunks 39 and 64.
undebug, used in chunk 64.

Exercises

Exercise 4.2

66 ⟨*permutations* 66⟩≡

```
(defun permutations (xs)
  (if (endp (cdr xs))
      (list xs)
      (loop for x in xs
            append (loop for ys in (permutations (remove x xs :count 1
                                                         :test  #'eq))
                         collect (cons x ys)))))
```

Defines:
permutations, never used.

Tests

67 ⟨*GPS Tests* 67⟩≡ (39)

```
(define-test complex
  (assert-equal
   (cons '(start)
         (mapcar #'(lambda (step) (list 'executing step))
                 '(look-up-number
                   telephone-shop
                   tell-shop-problem
                   give-shop-money
                   shop-installs-battery
                   drive-son-to-school)))
   (gps '(son-at-home car-needs-battery have-money have-phone-book)
        '(son-at-school)
        *school-ops*)))

(define-test unsolvable
  (assert-nil (gps '(son-at-home car-needs-battery have-money)
                   '(son-at-school)
                   *school-ops*)))

(define-test simple
  (assert-equal '((start) (executing drive-son-to-school))
                (gps '(son-at-home car-works)
                     '(son-at-school)
                     *school-ops*)))

(define-test money-leftover
  (assert-equal '((start) (executing drive-son-to-school))
                (gps '(son-at-home have-money car-works)
                     '(have-money son-at-school)
                     *school-ops*)))

(define-test clobbered-sibling
  (assert-nil (gps '(son-at-home car-needs-battery have-money have-phone-book)
                   '(have-money son-at-school)
                   *school-ops*)))
```

Uses `*school-ops*` 57.

ELIZA: *Dialog with a Machine*

68 ⟨*src/eliza.lisp* 68⟩≡

```
(in-package #:paip)
(defpackage #:paip.eliza
  (:use #:cl #:lisp-unit)
  (:export "eliza"))
(in-package #:paip.eliza)
```

⟨*ELIZA: Constants* 80⟩

⟨*ELIZA: Top-Level Function* 84⟩

⟨*ELIZA: Special Variables* 83⟩

⟨*ELIZA: Data Types* 82⟩

⟨*ELIZA: Functions* 85⟩

Uses `eliza` 84 and `use` 54.

ELIZA, one of the more well-known AI programs of the 1960s, simulates a psychotherapist, by way of a REPL.

Pattern Matching

69 ⟨*ELIZA: Pattern Matching* 69⟩≡ (85)

⟨*pat-match* 71⟩

⟨*match-variable* 70⟩

⟨*segment-match* 77⟩

⟨*segment-pattern-p* 78⟩

⟨*variable-p* 79⟩

ELIZA makes heavy use of pattern matching, which is at once versatile and limited.

Matching

70 ⟨*match-variable* 70⟩≡ (69)

```
(defun match-variable (var input bindings)
  "Does VAR match input? Uses (or updates) and returns bindings."
  (let ((binding (get-binding var bindings)))
    (cond ((not binding) (extend-bindings var input bindings))
          ((equal input (binding-val binding)) bindings)
          (t fail))))
```

Defines:
`match-variable`, used in chunks 73 and 77.
Uses `binding-val` 81, `extend-bindings` 81, `fail` 80, and `get-binding` 81.

Verify `var` is bound to `input` in `bindings`. If bound to another value, `fail`. If unbound, `extend-bindings`.

71 ⟨*pat-match* 71⟩≡ (69)

```
(defun pat-match (pattern input &optional (bindings no-bindings))
  "Match pattern against input in the context of the bindings."
  (cond ⟨Fail if the binding list is fail 72⟩
        ⟨Match a variable 73⟩
        ⟨If pattern equals input, return bindings 74⟩
        ⟨Match a segment 75⟩
        ⟨Call pat-match recursively 76⟩
        (t fail)))
```

Defines:
`pat-match`, used in chunks 76, 77, 80, and 86.
Uses `fail` 80 and `no-bindings` 80.

72 ⟨*Fail if the binding list is fail* 72⟩≡ (71)

```
((eq bindings fail) fail)
```

Uses `fail` 80.

If the binding list is `fail`, then the match fails, because some previous match must have failed.

73 ⟨*Match a variable* 73⟩≡ (71)

```
((variable-p pattern) (match-variable pattern input bindings))
```

Uses `match-variable` 70 and `variable-p` 79.

If the `pattern` is a single variable, return the result of `match-variable`; either `bindings` (possibly extended) or `fail`.

74 ⟨*If pattern equals input, return bindings* 74⟩≡ (71)

```
((eql pattern input) bindings)
```

If `pattern` equals `input`, return `bindings` as is.

75 ⟨*Match a segment* 75⟩≡ (71)

```
((segment-pattern-p pattern) (segment-match pattern input bindings))
```

Uses `segment-match` 77 and `segment-pattern-p` 78.

When both `pattern` and `input` are lists and the `(car pattern)` is a segment variable, match the variable to the initial part of the `input` and attempt to match `(cdr pattern)` to the rest.

76 ⟨*Call* `pat-match` *recursively* 76⟩≡ (71)

```
((and (consp pattern) (consp input))
 (pat-match (rest pattern) (rest input)
            (pat-match (first pattern) (first input)
                       bindings)))
```

Uses `pat-match` 71.

77 ⟨*segment-match* 77⟩≡ (69)

```
(defun segment-match (pattern input bindings &optional (start 0))
  "Match the segment pattern ((?* var) . pat) against input."
  (let ((var (second (first pattern)))
        (pat (rest pattern)))
    (if (null pat)
        (match-variable var input bindings)
        (let ((pos (position (first pat) input
                             :start start :test #'equal)))
          (if (null pos)
              fail
              (let ((b2 (pat-match pat (subseq input pos) bindings)))
                (if (eq b2 fail)
                    (segment-match pattern input bindings (+ pos 1))
                    (match-variable var (subseq input 0 pos) b2))))))))
```

Defines:
  `segment-match`, used in chunk 75.
Uses `fail` 80, `match-variable` 70, and `pat-match` 71.

Predicates

78 ⟨*segment-pattern-p* 78⟩≡ (69)

```
(defun segment-pattern-p (pattern)
  "Is this a segment matching pattern: ((?* var) . pat)"
  (and (consp pattern)
       (starts-with (first pattern) '?*)))
```

Defines:
  `segment-pattern-p`, used in chunk 75.
Uses `starts-with` 50.

79 ⟨*variable-p* 79⟩≡ (69)

```
(defun variable-p (x)
  "Is x a variable (a symbol beginning with '?')?"
  (and (symbolp x)
       (equal (char (symbol-name x) 0) #\?)))
```

Defines:
  `variable-p`, used in chunk 73.

Constants

80 ⟨*ELIZA: Constants* 80⟩≡ (68)

```
(defconstant fail nil
  "Indicates pat-match failure")

(defconstant no-bindings '((t . t))
  "Indicates pat-match success, with no variables.")
```

Defines:
  `fail`, used in chunks 70–72, 77, and 86.
  `no-bindings`, used in chunks 71 and 81.
Uses `pat-match` 71.

81 ⟨*ELIZA: Binding Functions* 81⟩≡ (85)

```
(defun get-binding (var bindings)
  "Find a (variable . value) pair in a binding list."
  (assoc var bindings))

(defun binding-val (binding)
  "Get the value part of a single binding."
  (cdr binding))

(defun lookup (var bindings)
  "Get the value part (for var) from a binding list."
  (binding-val (get-binding var bindings)))

(defun extend-bindings (var val bindings)
  "Add a (var . value) pair to a binding list."
  (cons (cons var val)
        (if (eq bindings no-bindings)
            nil
            bindings)))
```

Defines:
`binding-val`, used in chunk 70.
`extend-bindings`, used in chunk 70.
`get-binding`, used in chunk 70.
`lookup`, never used.
Uses `no-bindings` 80.

Rules

82 ⟨*ELIZA: Data Types* 82⟩≡ (68)

```
(defun rule-pattern (rule) (first rule))
(defun rule-responses (rule) (rest rule))
```

Defines:
`rule-pattern`, used in chunk 86.
`rule-responses`, used in chunk 86.

83 ⟨*ELIZA: Special Variables* 83⟩≡ (68)

```
(defparameter *eliza-rules*
  '((((?* ?x) hello (?* ?y))
     (How do you do.  Please state your problem.))
    (((?* ?x) I want (?* ?y))
     (What would it mean if you got ?y)
     (Why do you want ?y) (Suppose you got ?y soon))
    (((?* ?x) if (?* ?y))
     (Do you really think its likely that ?y) (Do you wish that ?y)
     (What do you think about ?y) (Really-- if ?y))
    (((?* ?x) no (?* ?y))
     (Why not?) (You are being a bit negative)
     (Are you saying "NO" just to be negative?))
    (((?* ?x) I was (?* ?y))
     (Were you really?) (Perhaps I already knew you were ?y)
     (Why do you tell me you were ?y now?))
    (((?* ?x) I feel (?* ?y))
     (Do you often feel ?y ?))
    (((?* ?x) I felt (?* ?y))
     (What other feelings do you have?))))
```

Eliza *Proper*

84 ⟨*ELIZA: Top-Level Function* 84⟩≡ (68)

```
(defun eliza ()
  "Respond to user input using pattern matching rules."
  (loop
    (print 'eliza>)
    (write (flatten (use-eliza-rules (read))) :pretty t)))
```

Defines:
  `eliza`, used in chunks 68 and 92.
Uses `flatten` 88 and `use-eliza-rules` 86.

85 ⟨*ELIZA: Functions* 85⟩≡ (68)

⟨*use-eliza-rules* 86⟩

⟨*switch-viewpoint* 87⟩

⟨*ELIZA: Pattern Matching* 69⟩

⟨*ELIZA: Binding Functions* 81⟩

⟨*flatten* 88⟩

⟨*random-elt* 37⟩

⟨*mappend* 36⟩

⟨*Is the argument a list that starts with a given atom?* 50⟩

86 ⟨*use-eliza-rules* 86⟩≡ (85)

```
(defun use-eliza-rules (input)
  "Find some rule with which to transform the input."
  (some #'(lambda (rule)
            (let ((result (pat-match (rule-pattern rule) input)))
              (unless (eq result fail)
                (sublis (switch-viewpoint result)
                        (random-elt (rule-responses rule))))))
        *eliza-rules*))
```

Defines:
`use-eliza-rules`, used in chunk 84.
Uses `fail` 80, `pat-match` 71, `random-elt` 37, `rule-pattern` 82, `rule-responses` 82, and `switch-viewpoint` 87.

87 ⟨*switch-viewpoint* 87⟩≡ (85)

```
(defun switch-viewpoint (words)
  "Change I to you and vice versa, and so on."
  (sublis '((I . you) (you . I) (me . you) (am . are))
          words))
```

Defines:
`switch-viewpoint`, used in chunk 86.

88 ⟨*flatten* 88⟩≡ (85)

```
(defun flatten (the-list)
  "Append together elements (or lists) in the list."
  (mappend #'mklist the-list))

(defun mklist (x)
  "Return x if it is a list, otherwise (x)."
  (if (listp x)
      x
      (list x)))
```

Defines:
`flatten`, used in chunk 84.
`mklist`, never used.
Uses `mappend` 36.

Build Software Tools

An Interactive Interpreter Tool

```
(defun program ()
  (loop
    (print prompt)
    (print (transform (read)))))
```

89 ⟨*interactive-interpreter* 89⟩≡ (91)

```
(defun interactive-interpreter (prompt transformer)
  "(`prompt' for and) read an expression, `transform' it and print the result."
  (loop
    (handler-case
        (progn
          (if (stringp prompt)
              (print prompt)
              (funcall prompt))
          (print (funcall transformer (read))))
      (error (condition)
        (format t "~&;; Error ~a ignored. Back to top level."
                condition)))))
```

⟨*prompt-generator* 90⟩

Defines:
`interactive-interpreter`, never used.

90 ⟨*prompt-generator* 90⟩≡ (89)

```
(defun prompt-generator (&optional (num 0) (ctl-string "[~d] "))
  "Return a function that prints prompts like [1], [2], etc."
  #'(lambda () (format t ctl-string (incf num))))
```

Defines:
`prompt-generator`, never used.

Package

91 ⟨*src/tools.lisp* 91⟩≡

```
(in-package #:paip)
(defpackage #:paip.tools
  (:use #:cl #:lisp-unit))
(in-package #:paip.tools)

⟨interactive-interpreter 89⟩
```

Uses use 54.

Package

92 ⟨*paip.asd* 92⟩≡

```
;;;; paip.asd

(asdf:defsystem #:paip
  :description "Paradigms of Artificial Intelligence Programming exercises"
  :author "Eric Bailey <eric@ericb.me>"
  :license "BSD-3"
  :depends-on (#:lisp-unit)
  :serial t
  :components ((:module "src"
                :serial t
                :components
                ((:file "intro")
                 (:file "gps")
                 (:file "eliza")
                 (:file "tools")))))

(defpackage #:paip
  (:use #:cl))
(in-package #:paip)
```

Uses eliza 84 and use 54.

Test Runner

93 ⟨*bin/runtests* 93⟩≡

```
#! /usr/bin/env bash

# N.B. quicklisp must be installed and configured.

sbcl --noinform --non-interactive \
     --userinit init.lisp \
     --eval "(in-package :paip.$1)" \
     --eval "(let* ((results  (lisp-unit:run-tests :all :paip.$1))
                    (failures (lisp-unit:failed-tests results))
                    (status   (if (null failures) 0 1)))
               (lisp-unit:print-failures results)
               (sb-posix:exit status))"
```

94 ⟨*init.lisp* 94⟩≡

```
#-quicklisp
(let ((quicklisp-init "quicklisp/setup.lisp"))
  (when (probe-file quicklisp-init)
    (load quicklisp-init)))

(push (concatenate 'string (sb-posix:getcwd) "/")
      asdf:*central-registry*)

(asdf:load-system :paip)
```

Chunks

⟨*Exercise 1.5 tests* 35⟩
⟨*Fail if the binding list is* `fail` 72⟩
⟨*find-all* 38⟩
⟨*flatten* 88⟩
⟨*function first-name(name):* 3⟩
⟨*GPS Tests* 67⟩
⟨*If* `pattern` *equals* `input`*, return* `bindings` 74⟩
⟨*if exp is an atom there is only one* 26⟩
⟨*if exp is nil there are if-null atoms* 25⟩
⟨*if item is equal to tree, there is one occurrence* 29⟩
⟨*if n is even return x to the power of n over two, squared* 20⟩
⟨*if n is zero return 1* 19⟩
⟨*if the first element of name is a title* 4⟩
⟨*if tree is an atom, there are no occurrences* 30⟩
⟨*init.lisp* 94⟩
⟨*interactive-interpreter* 89⟩
⟨*Is a condition an executing form?* 48⟩
⟨*Is the argument a list that starts with a given atom?* 50⟩
⟨*Is x an action?* 49⟩
⟨*last-name* 9⟩
⟨*mappend* 36⟩
⟨*Match a segment* 75⟩
⟨*Match a variable* 73⟩
⟨*match-variable* 70⟩
⟨*Maze Tests* 60⟩
⟨*Monkey and Bananas Example* 58⟩
⟨*Morton Downey, Jr.* 16⟩
⟨*multiply elements of the vectors pairwise and sum the results* 34⟩
⟨*Nursery School Example* 57⟩
⟨*otherwise add the count of the atoms in the first and rest of exp* 27⟩
⟨*otherwise return x times x to the power of n minus one* 21⟩
⟨*otherwise, add the occurrence within first the first and rest of tree* 31⟩
⟨*paip.asd* 92⟩
⟨*pat-match* 71⟩
⟨*permutations* 66⟩
⟨*power* 18⟩
⟨*Print debugging information* 65⟩
⟨*prompt-generator* 90⟩
⟨*random-elt* 37⟩
⟨*Rex Morgan MD* 15⟩
⟨*segment-match* 77⟩
⟨*segment-pattern-p* 78⟩
⟨*Solve a goal from a state using a list of operators* 42⟩
⟨*square* 22⟩

⟨*src/eliza.lisp* 68⟩
⟨*src/gps.lisp* 39⟩
⟨*src/intro.lisp* 7⟩
⟨*src/tools.lisp* 91⟩
⟨*suffixes* 13⟩
⟨*switch-viewpoint* 87⟩
⟨*Test if an element is equal to a member of a list* 55⟩
⟨*The Blocks World Domain* 61⟩
⟨*the last element of a name is a suffix* 10⟩
⟨*The Maze Searching Domain* 59⟩
⟨*then return the* `last-name` *of all but the last element of the name* 11⟩
⟨*then return the first-name of the rest of the name* 5⟩
⟨*titles* 1⟩
⟨*Use a list of of operators* 54⟩
⟨*use-eliza-rules* 86⟩
⟨*variable-p* 79⟩

Index

Bibliography